James Webb Space Telescope (JWST)

Mid-Infrared Instrument (MIRI)

Photon Conversion Efficiencies

Prepared by:	Alistair Glasse	Date:	19th October 2016
Approved by:		Date:	

Change Log

Issue		Date
1	First issue	13/11/14
2	Minor changes to ImPCE.xls (Issue 2) to fix column labels.	29/5/15
3	Clarification of intended usage (p4)	27/8/15
4	(Another) minor change to column labels in ImPCE (Iss 3)	2/12/15
5	Conversion of files to fits format for use by Exposure Time Calculator. LRS sampling resolution increased.	22/6/16
6	Included coronagraph PCEs	19/10/16

PAGE LEFT INTENTIONALLY BLANK

Contents

References

Reference ID	Document ID	Title	Issue	Date
RD-1	MIRI-TN-00061	Sensitivity Modelling of the MIRI Flight Model	3	Dec 13
RD-2	Glasse et al. PASP, 2015	Mid-Infrared Instrument for the James Webb Space Telescope IX: Predicted Sensitivity	-	Aug 15
RD-3	JWST-STScI-00XXXX SM-12	Format and Delivery Procedures for JWST ETC Reference Files, Gilbert, K., Laidler, V., Pontoppidan, K., Pickering, T.	-	~ 2014
RD-4	CDP Meta Data	http://miri.ster.kuleuven.be/bin/ view/Internal/Software/MiriCalfileMetaData	-	9/6/16
RD-5	PASP, v127, p633	The Mid-Infrared Instrument for the James Webb Space Telescope, V: Predicted Performance of the MIRI Coronagraphs (Boccaletti et al.)	-	2015

1 Introduction

This technical note acts as a wrapper for the control and delivery of the Photon Conversion Efficiency values used in the MIRI sensitivity model [RD-1] and whose results are published in [RD-2].

From Issue 5 the data are provided in a format suitable for use in the STScI Exposure Time Calculator (ETC) as delivered by the MIRI Team's Calibration Data Product mechanism. The data values are unchanged compared to previous issues, apart from the LRS, where the number of wavelength samples has been increased and PCE values for SLITLESS LRS spectroscopy are now included.

These reference files are being delivered as part of the MIRI Calibration Data Product mechanism (CDP6), so any requests for corrections or format changes can be requested via the CDP Delivery Review Board, in which STScI is represented by Dean Hines.

2 Description of File Contents

The PCE is equal to the number of electrons detected per photon incident on the MIRI entrance focal plane within the nominal science beam (ie they exclude the telescope optics).

They are derived from the product of

- The detector Quantum Efficiency (Curves provided by Ressler, private communication).
- Mirror reflectivities (0.98 per surface, based on measurements)
- A wavelength independent transmission factor of 0.8 which represents the Beginning of Life (BOL) contamination.

Note 1. The figures DO NOT include a factor for the End Of Life (EOL) contamination. This factor IS included (set equal to 0.8) in the sensitivity model (RD-2).

Note 2. The nominal transmission of the JWST OTE, which is assumed to be 0.88 at start of mission in RD-2, is NOT included in the PCE curves.

The filter profiles have been clipped in wavelength to remove measurement artefacts. In the case of the F2550W and F2550WR filters, the profiles are truncated at a PCE of ~ 0.74 % at λ = 30 micron by the measurement range of a sub-component. For λ > 30 micron we recommend using a linear extrapolation of the form,

$$PCE(\lambda \geq 30\ \mu m) = PCE(\lambda = 30\ \mu m) - ((\lambda - 30\ \mu m) / 100),$$

with negative PCE values set equal to zero.

For the coronagraphs, the PCE applies to positions well away from the null in the 4 quadrant phase masks, with the transmission of the pupil mask mounted with each coronagraphic filter in the filter wheel taken as 0.62 for the 4QPMs and 0.72 for the Lyot mask (27 (RD-5). We assign a transmission 0.9 for the 4QPMs, with the resulting peak PCE (plotted in Figure X) falling in the range 0.15 to 0.20, in accordance with RD-5.

3 Format

The PCEs are provided as table extensions in 'fits' data files. The format of a typical reference file header is shown in Figure 1. The metadata and parameter names have been chosen to match the CDP standard (RD-4), whilst also following as closely as possible the convention set out for ETC reference files in RD-3. We note that the ETC convention does not describe PCEs very well, but private communication with Nick Earl, who is the person actually implementing the PCE curves in the ETC, suggests that the format provided is satisfactory.

```
SIMPLE  =                    T /image conforms to FITS standard
BITPIX  =                   16 /array data type
NAXIS   =                    0 /number of array dimensions
EXTEND  =                    T /file may contain extensions
TELESCOP= 'JWST    '           /The telescope used to acquire the data
SYSTEM  = 'MODELLING'          /Data intended for use in ETC
INSTRUME= 'MIRI    '           /Instrument used to acquire data
DETECTOR= 'MIRIFUSHORT'        /Name of detector
READPATT= 'ANY     '           /Detector readout pattern
SUBSTRT1=                    1 /
SUBSIZE1=                 1032 /
SUBSTRT2=                    1 /
SUBSIZE2=                 1024 /
VERSION = '06.00.00'           /
PEDIGREE= 'GROUND  '           /
MODELNAM= 'FM      '           /Instrument used to acquire data
DETSETNG= 'ANY     '           /Applicable to any detector clock/bias settings
FILETYPE= 'DATA TABLE'         /ETC data file type
FILENAME= 'MIRI_FM_MIRIFUSHORT_2MEDIUM_PCE_06.00.00.fits' /CDP Reference file na
ETCNAME = 'jwst_miri_mirifushort_2medium_pce.fits' /ETC file name
DATE    = '2016-06-22T12:41:33.00' /
USEAFTER= '2000-01-01T00:00:00.00' /
REFTYPE = 'PCE     '           /Type of data found in file
COMPNAME= 'MIRIFUSHORT_2MEDIUM' /Name of component
AUTHOR  = 'Alistair Glasse'    /Author of CDP
ORIGIN  = 'MIRI European Consortium' /The organization responsible
DESCRIP = 'MRS Photon Conversion Efficiency Profiles' /Summary of content
LITREF  = 'Glasse et al., PASP v127, (2015) Fig. 3' /Literature reference
HISTORY  Description of reference file creation
HISTORY DOCUMENT: MIRI-TN-00072-ATC Iss 5 - PCE Report
HISTORY REF1 MIRI-TN-00020-ATC Iss 2, Spectrometer Filter Transmission (OBA EIDP
HISTORY REF2 MIRI DFM-289 04.02, Predicted DQE of the MIRI FM SCAs (FPS EIDP)
HISTORY SOFTWARE: IDL 8.1 MIRI_ETC_PCE.pro, (MIRI-TN-00077-ATC)
HISTORY DATA USED: Sub-system measurements plus modelling.
HISTORY DIFFERENCES: N/A
CHANNEL = '2       '           /MRS channel
BAND    = 'MEDIUM  '           /MRS spectral band
SUBARRAY= 'GENERIC '           /Detector subarray
EXP_TYPE= 'MIR_MRS '           /Exposure type
END
```

Figure 1 Header block for MIRI_FM_CHANNEL2MEDIUM_PCE_06.00.00.fits

4 PCE Plots

For convenience and to help with user's own version control we provide plots of the data below. We may update future issues of this note to discuss the origin of the provided PCE values, but the reader is referred to RD-2 in the first instance.

4.1 Imager



Figure 2 Plot of Imager PCE profiles.

4.2 Coronagraphs



Figure 3 Plot of Coronagraph PCE profiles.

4.3 Low Resolution Spectrometer



Figure 4 Plot of LRS PCEs for the enslitted and slitless configurations. Note that the feature in the slitless curve around 4 microns may be a measurement artifact.

4.4 Medium Resolution Spectrometer



Figure 5 Plot of MRS PCE data points.

5 A Worked Example

We will now demonstrate how the PCE curves can be used in practice, taking as an example an observation using the F1000W filter of an extended target with a spectral energy distribution given by

$\nu F_\nu(\lambda)$ = Constant,

where $F_\nu(\lambda)$ is the flux per unit frequency interval at a wavelength 'λ'. We assume that the target flux at λ = 10 µm is F_ν(10 µm) = 0.5 millJansky per square arcsecond.

Taking a JWST collecting area of $A_{tel} = 25\ m^2$, the transmission for the mirrors between the target and the MIRI entrance focal plane to be $\tau_{OTE} = 0.88$, and the field of view of an imager pixel as $\Omega_{pixel} = (0.11\ arcsec)^2$, we can express the photon flux per pixel at the MIRI entrance focal plane in wavelength density units as,

$P_\lambda(\lambda) = F_\nu(\lambda)\ A_{tel}\ /\ (h\ \lambda)\ (arcsec^2\ /\ \Omega_{pixel})$,

with $P_\lambda(10\ \mu m) = 228.3$ photon $sec^{-1}\ \mu m^{-1}\ pixel^{-1}$.

This conversion to a wavelength density simplifies the process of integration over wavelength, since the PCE curve is provided on a uniform wavelength grid. In fact, our choice of $\nu F_\nu(\lambda)$ = C has the convenient property of giving a $P_\lambda(\lambda)$ variation which is constant. We then illustrate the numerical integration step to arrive at a final in-band pixel signal in the table below, where a very coarse sampling grid of $\Delta\lambda$ = 0.25 µm is used to aid readability.

Wavelength, λ	PCE(λ)	PCE(λ) x $P_{\lambda}(\lambda)$ x Δλ
micron	el. photon^{-1}	el. sec-1 pixel-1
8.9	0.009	0.8
9.3	0.384	35.1
9.7	0.390	35.6
10.1	0.399	36.4
10.5	0.415	37.9
10.9	0.189	17.3
11.3	0.000	0.0
Integrated signal in F1000W band		**163.1**

The estimated signal in the F1000W filter for our 0.5 mJy target is then estimated to be 163.1 electrons sec^{-1} pixel^{-1}.

6 Code Listing of MIRI_ETC_PCE

```
;=====================================================================================
;IDL 8.1 programme to generate ETC reference files describing the components of the
;  MIRI Imager Photon Conversion Efficiency, including filter transmissions, detector
;  QE etc.
;
;  Author - A Glasse, ()
;  Version - 1.0  (7/6/16) - First Version
;  Version - 2.0  (19/10/16) - Added Coronagraph PCEs.
pro MIRI_ETC_PCE

  inPath = 'F:\MIRI\Sensitivity\Flight Model\Issue6\'
  outPath = 'F:\MIRI\ETC\Ref_Files\'

  filterFileStub = 'MIRI-TN-00068_'

  filterDataFile = inPath + filterFileStub + 'ImPCE.csv'
  MIRI_ETC_Imager_PCE, filterDataFile, outPath

  filterDataFile = inPath + filterFileStub + 'CorPCE.csv'
  MIRI_ETC_Coronagraph_PCE, filterDataFile, outPath

  filterDataFile = inPath + filterFileStub + 'LRSPCE.csv'
  MIRI_ETC_LRS_PCE, filterDataFile, outPath

  filterDataFile = inPath + filterFileStub + 'MRSPCE.csv'
  MIRI_ETC_MRS_PCE, filterDataFile, outPath

  print, 'Done'
end
;-------------------------------------------------------------------------------------
pro MIRI_ETC_Coronagraph_PCE, filterDataFile, outPath

eidpRef = ['MIRI-RP-00012-UOS Iss 4, Imager filter deliveries (OBA EIDP)', $
           'MIRI DFM-289 04.02, Predicted DQE of the FM SCAs (FPS EIDP)']
description = 'Coronagraph filter passband Photon Conversion Efficiencies (PCE)'

dataBlock = read_csv(filterDataFile, HEADER = headerRow, N_TABLE_HEADER = 1)

w = dataBlock.(0)

nFilters = size(headerRow, /N_ELEMENTS) - 1
nWavePoints = size(w, /N_ELEMENTS)

for i = 0, nFilters - 1 do begin
```

```
  compUC = headerRow[i+1]
  detector = 'MIRIMAGE'
  cdpFileName = 'MIRI_FM_' + detector + '_' + compUC + '_PCE_7B.01.00.fits'

  createPrimaryHeader, cdpFileName, pHeader, compUC, eidpRef, description, detector
  sxaddpar, pHeader, 'FILTER', compUC, 'Coronagraph filter name'    ;Add specific
keywords
  sxaddpar, pHeader, 'SUBARRAY', 'GENERIC', 'Detector subarray'
  sxaddpar, pHeader, 'EXP_TYPE', 'MIR_IMAGE', 'Exposure type'

  outFile = outPath + cdpFileName
  MWRFITS, 0, outFile, pHeader, /CREATE

  t = dataBlock.(i+1)
  createWavPCETable, w, t, wtTable, wtHeader
  MWRFITS, wtTable, outFile, wtHeader

endfor
end

;-------------------------------------------------------------------------------------
pro MIRI_ETC_Imager_PCE, filterDataFile, outPath

eidpRef = ['MIRI-RP-00012-UOS Iss 4, Imager filter deliveries (OBA EIDP)', $
           'MIRI DFM-289 04.02, Predicted DQE of the FM SCAs (FPS EIDP)']
description = 'Imager filter passband Photon Conversion Efficiencies (PCE)'

dataBlock = read_csv(filterDataFile, HEADER = headerRow, N_TABLE_HEADER = 1)

w = dataBlock.(0)

nFilters = size(headerRow, /N_ELEMENTS) - 1
nWavePoints = size(w, /N_ELEMENTS)

for i = 0, nFilters - 1 do begin

  compUC = headerRow[i+1]
  detector = 'MIRIMAGE'
  cdpFileName = 'MIRI_FM_' + detector + '_' + compUC + '_PCE_7B.01.00.fits'

  createPrimaryHeader, cdpFileName, pHeader, compUC, eidpRef, description, detector
  sxaddpar, pHeader, 'FILTER', compUC, 'Imager filter name'    ;Add specific keywords
  sxaddpar, pHeader, 'SUBARRAY', 'GENERIC', 'Detector subarray'
  sxaddpar, pHeader, 'EXP_TYPE', 'MIR_IMAGE', 'Exposure type'

  outFile = outPath + cdpFileName
  MWRFITS, 0, outFile, pHeader, /CREATE

  t = dataBlock.(i+1)
  createWavPCETable, w, t, wtTable, wtHeader
  MWRFITS, wtTable, outFile, wtHeader

endfor
end
;-------------------------------------------------------------------------------------
pro MIRI_ETC_LRS_PCE, filterDataFile, outPath

;Path to raw data (PCE and transmissions read by Mathcad sensitivity model
eidpRef = ['MIRI-TR-10013-MPI Iss 1.7, LRS Spectrophotometric Perf. (OBA EIDP)', $
           'MIRI DFM-289 04.02, Predicted DQE of the MIRI FM SCAs (FPS EIDP)']
description = 'LRS Photon Conversion Efficiency Profiles'
slitName = ['P750L', 'P750L_SLITLESSPRISM']
subArray = ['GENERIC', 'SLITLESSPRISM']

dataBlock = read_csv(filterDataFile, HEADER = header, N_TABLE_HEADER = 0)

w = dataBlock.(0)

nSlitConfigs = 2
```

```
for i = 0, nSlitConfigs - 1 do begin

  detector = 'MIRIMAGE'
  compUC = slitName[i]
  cdpFileName = 'MIRI_FM_' + detector + '_' + compUC + '_PCE_06.00.00.fits'
  createPrimaryHeader, cdpFileName, header, compUC, eidpRef, description, detector
  sxaddpar, header, 'FILTER', 'P750L', 'LRS prism'
  sxaddpar, header, 'SUBARRAY', subArray[i], 'Imager detector subarray'
  sxaddpar, header, 'EXP_TYPE', 'MIR_LRS-FIXEDSLIT', 'Exposure type'

  if (i eq 1) then begin
    sxaddpar, header, 'SUBSIZE1', 72
    sxaddpar, header, 'SUBSTRT2', 348
    sxaddpar, header, 'SUBSIZE2', 416
    sxaddpar, header, 'EXP_TYPE', 'MIR_LRS-SLITLESS', 'Exposure type'
  endif

  outFile = outPath + cdpFileName
  MWRFITS, 0, outFile, header, /CREATE

  pce = dataBlock.(i+1)
  createWavPCETable, w, pce, wtTable, wtHeader
  MWRFITS, wtTable, outFile, wtHeader

endfor

end
;-------------------------------------------------------------------------------------
pro MIRI_ETC_MRS_PCE, filterDataFile, outPath

eidpRef = ['MIRI-TN-00020-ATC Iss 2, Spectrometer Filter Transmission (OBA EIDP)', $
           'MIRI DFM-289 04.02, Predicted DQE of the MIRI FM SCAs (FPS EIDP)']
description = 'MRS Photon Conversion Efficiency Profiles'

detectorName = ['MIRIFUSHORT', 'MIRIFUSHORT', 'MIRIFULONG', 'MIRIFULONG']
channelName = ['1', '2', '3', '4']
bandName = ['SHORT', 'MEDIUM', 'LONG']

dataBlock = read_csv(filterDataFile, HEADER = header, N_TABLE_HEADER = 0)

band = dataBlock.(0)
w = dataBlock.(1)
pce = dataBlock.(2)

nGratingPositions = 3
nSubBands = 12

nRecords = size(band, /N_ELEMENTS)

nRec_Band = nRecords / nSubBands

rStart = 0
for i = 0, nSubBands - 1 do begin
  rEnd = rStart + nRec_Band - 1
  chan = (i / nGratingPositions)
  subBand = i mod nGratingPositions
  detector = detectorName[chan]
  compUC = detector + '_' + channelName[chan] + bandName[subBand]
  cdpFileName = 'MIRI_FM_' + compUC + '_PCE_06.00.00.fits'

  createPrimaryHeader, cdpFileName, header, compUC, eidpRef, description, detector
  sxaddpar, header, 'CHANNEL', channelName[chan], 'MRS channel'
  sxaddpar, header, 'BAND', bandName[subBand], 'MRS spectral band'
  sxaddpar, header, 'SUBARRAY', 'GENERIC', 'Detector subarray'
  sxaddpar, header, 'EXP_TYPE', 'MIR_MRS', 'Exposure type'

  outFile = outPath + cdpFileName
  MWRFITS, 0, outFile, header, /CREATE

  createWavPCETable, w[rStart:rEnd], pce[rStart:rEnd], wtTable, wtHeader
```

```
  MWRFITS, wtTable, outFile, wtHeader

  rStart = rEnd + 1
endfor

end
;==================================================================================
pro createPrimaryHeader, fileName, header, compUC, eidpRef, description, detector

fileType = 'DATA TABLE'
refType = 'PCE'
litRef = 'Glasse et al., PASP v127, (2015) Fig. 3'

compLC = STRLOWCASE(compUC)

etcFileName = 'jwst_miri_' + compLC + '_pce' + '.fits'

header = !NULL    ;Create header from scratch every time
sxaddpar, header, 'SIMPLE', 'T', 'image conforms to FITS standard'
sxaddpar, header, 'BITPIX', 16, 'array data type'
sxaddpar, header, 'NAXIS', 2, 'number of array dimensions'
sxaddpar, header, 'EXTEND', 'T', 'file may contain extensions'
sxaddpar, header, 'TELESCOP', 'JWST', 'The telescope used to acquire the data'
sxaddpar, header, 'SYSTEM', 'MODELLING', 'Data intended for use in ETC'
sxaddpar, header, 'INSTRUME', 'MIRI', 'Instrument used to acquire data'
sxaddpar, header, 'DETECTOR', detector, 'Name of detector'
sxaddpar, header, 'READPATT', 'ANY', 'Detector readout pattern'
sxaddpar, header, 'SUBSTRT1', 1
sxaddpar, header, 'SUBSIZE1', 1032
sxaddpar, header, 'SUBSTRT2', 1
sxaddpar, header, 'SUBSIZE2', 1024
sxaddpar, header, 'VERSION', '06.00.00'
sxaddpar, header, 'PEDIGREE', 'GROUND'

sxaddpar, header, 'MODELNAM', 'FM', 'Instrument used to acquire data'
sxaddpar, header, 'DETSETNG', 'ANY', 'Applicable to any detector clock/bias settings'
sxaddpar, header, 'FILETYPE', fileType, 'ETC data file type'
sxaddpar, header, 'FILENAME', fileName, 'CDP Reference file name'
sxaddpar, header, 'ETCNAME', etcFileName, 'ETC file name'
julian = systime(/JULIAN, /UTC)
dateString = date_conv(julian, 'F')
sxaddpar, header, 'DATE', dateString
useafter = julday(1, 1, 2000, 0, 0, 0)
useafterString = date_conv(useafter, 'F')
sxaddpar, header, 'USEAFTER', useafterString
sxaddpar, header, 'REFTYPE',  refType, 'Type of data found in file'
sxaddpar, header, 'COMPNAME', compUC, 'Name of component'

sxaddpar, header, 'AUTHOR  ', 'Alistair Glasse' , 'Author of CDP'
sxaddpar, header, 'ORIGIN',    'MIRI European Consortium', $
                               'The organization responsible'
sxaddpar, header, 'DESCRIP', description, 'Summary of content'
sxaddpar, header, 'LITREF', litRef, 'Literature reference'
sxaddpar, header, 'HISTORY', ' Description of reference file creation'
sxaddpar, header, 'HISTORY', 'DOCUMENT: MIRI-TN-00072-ATC Iss 5 - PCE Report'
sxaddpar, header, 'HISTORY', 'REF1 ' + eidpRef[0]
sxaddpar, header, 'HISTORY', 'REF2 ' + eidpRef[1]
sxaddpar, header, 'HISTORY', 'SOFTWARE: IDL 8.1 MIRI_ETC_PCE.pro, (MIRI-TN-00077-ATC)'
sxaddpar, header, 'HISTORY', 'DATA USED: Sub-system measurements plus modelling.'
sxaddpar, header, 'HISTORY', 'DIFFERENCES: N/A'

end
;----------------------------------------------------------------------------------
pro createWavPCETable, w, t, table, header

nPoints = size(w, /N_ELEMENTS)

table = [{WAVELENGTH:0.0, EFFICIENCY:0.0, CONVERSION:0.0}]
for i = 0, nPoints - 1 do begin
  table = [table, $
```

```
    {WAVELENGTH:float(w[i]), EFFICIENCY:float(t[i]), CONVERSION:float(1.0)}]
endfor
table = table[1:*]      ;remove dummy first row

;We are writing 2 float values so nBytes = 8
nBytes = 8
dummyByteArray = bytarr(nBytes, nPoints)
header = createImageHeader(dummyByteArray, 'DATA_TABLE')
addColumnHeader, header, '1', 'WAVELENGTH', 'F', 'microns     '
addColumnHeader, header, '2', 'EFFICIENCY', 'F', 'fraction    '
addColumnHeader, header, '3', 'CONVERSION', 'F', 'el / photon'

end
;-------------------------------------------------------------------------------------
function createImageHeader, image, extName

  imSize = size(image)
  type = size(image, /TYPE)
  nAxes = imSize[0]
  nX = imSize[1]
  nY = imSize[2]
  if (nAxes gt 2) then nZ = imSize[3] else nZ = 0
  if (type eq 1) then bitpix = 8
  if (type eq 2) then bitpix = 16
  if (type eq 3) then bitpix = 32
  if (type eq 4) then bitpix = -32
  if (type eq 5) then bitpix = -64

  sxaddpar, header, 'XTENSION', 'IMAGE', 'Image extension'
  sxaddpar, header, 'BITPIX', bitpix, ' Array data type'
  sxaddpar, header, 'NAXIS', nAxes, 'Number of data dimensions'
  sxaddpar, header, 'NAXIS1', nX, ' '
  sxaddpar, header, 'NAXIS2', nY, ' '
  if (nZ gt 0) then sxaddpar, header, 'NAXIS3', nZ, ' '
  sxaddpar, header, 'PCOUNT', 0, 'Number of parameters'
  sxaddpar, header, 'GCOUNT', 1, 'Number of groups'
  sxaddpar, header, 'EXTNAME ', extName, 'Extension name'
  sxaddpar, header, 'BUNIT ', ' ', 'Data unit'

  return, header
end
;-------------------------------------------------------------------------------------
pro addColumnHeader, header, column, type, form, unit

  sxaddpar, header, 'TTYPE' + column, type
  sxaddpar, header, 'TFORM' + column, form
  if (N_Elements(unit) ne 0) then sxaddpar, header, 'TUNIT' + column, unit
end
;=====================================================================================
```

A Glasse, 19/10/16